テスト駆動開発でお試しする Pharo Smalltalk・第2回 等価性チェックと乗算の実装

[新訳版『テスト駆動開発』](https://estore.ohmsha.co.jp/titles/978427421788P)の第Ⅰ部の多国通貨をネタに Smalltalk を学ぶ[シリーズ]()の第2回です。同書と似た仕様を目指しますが、好みを優先する等の都合で写経にはなっていませんのでどうぞあしからず。（12/24追記：なるべく忠実な写経は別に書きました→[こちら](https://qiita.com/sumim/items/791984a4702bce91dacd) ）


- [テスト駆動開発でお試しする Pharo Smalltalk・目次](https://qiita.com/sumim/items/3a36c03ab41519077b37)

 [前回](https://qiita.com/sumim/items/fa41066c57d211814ff9) は 5 USD というメッセージ式が Dollar のインスタンスを返すところまでを実装しました。

今回はこの Dallar のインスタンスをもう少しまともなものにしてゆきます。
<br />
#####*参考： 多国通貨実装のおおまかな流れ*
1. **Dollar を定義** ← 今ここ
1. Dollar をコピーして Franc を定義
1. Dollar、Franc の重複を新しく作ったスーパークラスの Money にプルアップ
1. Sum（Expression）、Bank を定義して reduce を実現

# Dollar の等価性
Dollar にインスタンス変数を追加して状態を持たせ、インスタンス間で等価（および、不等価）をチェックできるようにしましょう。

###### **TODO リスト**
- [ ] 5 USD + 10 CHF = 10 USD （レートが 2 : 1 の場合）
- [ ] 5 USD * 2 = 10 USD
- [x] 5 USD が Dollar を返す
- [ ] **5 USD = 5 USD**
- [ ] **5 USD ~= 6 USD**

まずは等価性のテストを書きます。

```Smalltalk
MoneyTest >> testEquality
	self assert: 5 USD = 5 USD
```

MoneyTest の ○印をクリックしてテストを実行すると今回はレッド（エラー）にはならずイエロー（失敗）になります。

![780.png](https://qiita-image-store.s3.amazonaws.com/0/49275/34279e24-1bd8-915b-1ba8-e000c3f60e20.png)


等価性チェックを行なう = メソッドは（ProtoObject を除く）全オブジェクトのスーパークラスである Object に定義済みで、デフォルトの振る舞いでは同一オブジェクトかどうかを返します。このため、先の testEquality はエラーにはなりませんでした。

```Smalltalk
Object >> = anObject 
	"Answer whether the receiver and the argument represent the same 
	object. If = is redefined in any subclass, consider also redefining the 
	message hash."

	^self == anObject 
```

しかし、最初の 5 USD と二番目の 5 USD は個々に生成された別のオブジェクトであるため同一にはならずテストには失敗したわけです。

仮実装（最短のグリーン化）は省いて、ステップを大きく進めましょう。

###### **TODO リスト**
- [ ] 5 USD + 10 CHF = 10 USD （レートが 2 : 1 の場合）
- [ ] 5 USD * 2 = 10 USD
- [x] 5 USD が Dollar を返す
- [ ] **5 USD = 5 USD**
- [ ] **5 USD ~= 6 USD**
- [ ] インスタンス変数 amount を定義
- [ ] amount のアクセッサーメソッド（amount 、amount: ） を追加

まず、不等価も念のためテストに追加しておきます。

```Smalltalk
MoneyTest >> testEquality
	self assert: 5 USD = 5 USD.
	self assert: 5 USD ~= 6 USD
```

ちなみに ~= は不等価チェックです。デフォルト実装は次のコードになっていて、テンプレートメソッド・パターン（[デパ](https://www.amazon.co.jp/dp/4797311126)p347）により、= のみの再定義で機能します。

```Smalltalk
Object >> ~= anObject 
	"Answer whether the receiver and the argument do not represent the 
	same object."

	^self = anObject == false
```

インスタンス変数は、クラス定義の instanceVariableNames: '' の引数っぽくなっている文字列に（複数あるときはスペースで区切って）指定すればいいのですが、今回は機能の紹介を兼ねて Pharo のリファクタリング機能を使ってみましょう。

Dollar クラスを右クリック → Refactoring → Inst Var Refactoring →　 Add を選びます。

![781.png](https://qiita-image-store.s3.amazonaws.com/0/49275/e091e9f8-d8a8-5ea6-b8e1-0c8a63dd7739.png)

Provide the following information で追加したいインスタンス名を求められるので、inst を消して amount に変え OK します。

![Provide the following information.png](https://qiita-image-store.s3.amazonaws.com/0/49275/bb31d106-ad0c-973b-9626-a60c2e708f77.png)

すると（今回は amount の追加だけなので項目はひとつしか出ませんが）リファクタリングに伴う作業の一覧が表示され、本当に実行してよいか確認を促されるのでそのまま OK します。

![Changes Browser.png](https://qiita-image-store.s3.amazonaws.com/0/49275/f46bc24a-f98c-d2d4-2db5-a5cdff88d6a8.png)

Dollar を二度クリックして再度選択する必要がありますが、先述の instanceVariableNames: の引数が '' から 'amount' に変わっているのが確認できるはずです。

![783.png](https://qiita-image-store.s3.amazonaws.com/0/49275/e95527a4-aaab-fbe6-ad13-bb2aca260e8f.png)

Smalltalk のインスタンス変数は Java などのそれとは違い、インスタンスメソッド外から直接アクセスする手段がありません。インスタンス外から値を読み取ったり設定する必要があるときは、通常は、あらかじめアクセッサーメソッドを定義しておき、それらを介する必要があります（アクセッサーメソッドなんかなくとも Smalltalk の強力なリフレクション機能を使って済ませる手もありますが、それはまた別の話）。

Pharo では、インスタンス変数の追加したときと同じ Refactoring メニューにインスタンス変数のアクセッサーを自動生成する機能もあるのでこれを使いましょう。

Dollar クラスを右クリック → Refactoring → Inst Var Refactoring → Accessors を選びます。

![787.png](https://qiita-image-store.s3.amazonaws.com/0/49275/fcb9fcf7-f2ef-eb20-692b-7e44508133bd.png)

アクセッサーを自動生成したいインスタンス変数を指示するように促されるので amount をクリックして選択します。

![788.png](https://qiita-image-store.s3.amazonaws.com/0/49275/d41f4d7e-55cc-6fb6-5adb-d9458890d2a4.png)

続けて、ゲッター（インスタンス変数と同名のメソッド）、セッター（インスタンス変数に : を付けた引数をひとつとるメソッド）、どちらのメソッドを生成するかを尋ねてくるので、そのまま OK します。

accessing プロトコルが新たに設けられ、そこに amount 、amount: が定義されます。メソッドの中身の確認ついでに、amount: の仮引数が anObject なのがあんまりなので、aNumber とでもしておきましょう。

```Smalltalk
Dollar >> amount: aNumber
	amount := aNumber
```

![791.png](https://qiita-image-store.s3.amazonaws.com/0/49275/e37218d4-578f-72da-0720-3c6e7868c5e3.png)

###### **TODO リスト**
- [ ] 5 USD + 10 CHF = 10 USD （レートが 2 : 1 の場合）
- [ ] 5 USD * 2 = 10 USD
- [x] 5 USD が Dollar を返す
- [ ] **5 USD = 5 USD**
- [ ] **5 USD ~= 6 USD**
- [x] インスタンス変数 amount を定義
- [x] amount のアクセッサーメソッド（amount 、amount: ） を追加

準備は整ったので改めて Dollar オブジェクトの等価性チェック Dollar>> = を定義しまます。Dollar クラスのプロトコルリスト枠にある -- all -- をクリックして、下のコードペインに（Dallar クラスの定義式ではなく）メソッド定義用テンプレートが現れたのを確認して次のように = メソッドを定義します。

```Smalltalk
Dollar >> = other
	^ other class == self class and: [ other amount = amount ]
```

プロトコルは as yet classified に分類されるので、同枠内で右クリック → Categorize all uncategorized を選んで、comparing に自動割り当てしてもらいます。

![793.png](https://qiita-image-store.s3.amazonaws.com/0/49275/a3673179-8e70-f514-9635-5c889e0caf62.png)

テストは相変わらず失敗するはずですが、レッドになってしまっていないことは確認しておきます。

![796.png](https://qiita-image-store.s3.amazonaws.com/0/49275/19c7cf5c-72e6-506c-7c2b-6170eac4f371.png)

最後に、Number>>USD が amount をセットした適切な Dallar を返すように修正します。

```Smalltalk
Number >> USD
	^ Dollar new amount: self; yourself
```

![797.png](https://qiita-image-store.s3.amazonaws.com/0/49275/3732a418-bdf3-880e-b828-a3e1b136e102.png)

「 ; yourself 」というのは、直前の式の結果を無視してレシーバーを返す「カスケード式」と呼ばれる“おまじない”（シンタックスシュガー）です。次のように書くのと意味は同じです。

```Smalltalk
Number >> USD
	| aDollar |
	aDollar := Dollar new.
	aDollar amount: self.
	^ aDollar
```

これでテストは通るはずです。

![798.png](https://qiita-image-store.s3.amazonaws.com/0/49275/0bc19266-ff09-d557-4da3-d08b83e9e95b.png)

###### **TODO リスト**
- [ ] 5 USD + 10 CHF = 10 USD （レートが 2 : 1 の場合）
- [ ] 5 USD * 2 = 10 USD
- [x] 5 USD が Dollar を返す
- [x] **5 USD = 5 USD**
- [x] **5 USD ~= 6 USD**
- [x] インスタンス変数 amount を定義
- [x] amount のアクセッサーメソッド（amount 、amount: ） を追加

# Dollar の乗算
ずいぶん大股歩きをしてしまいましたが、最後にさくっと乗算のサポートをしておしまいにしましょう。

###### **TODO リスト**
- [ ] 5 USD + 10 CHF = 10 USD （レートが 2 : 1 の場合）
- [ ] **5 USD * 2 = 10 USD**
- [x] 5 USD が Dollar を返す
- [x] 5 USD = 5 USD
- [x] 5 USD ~= 6 USD
- [x] インスタンス変数 amount を定義
- [x] amount のアクセッサーメソッド（amount 、amount: ） を追加

テストを書きます。

```Smalltalk
MoneyTest >> testMultiplication
	self assert: 5 USD * 2 = 10 USD
```

![799.png](https://qiita-image-store.s3.amazonaws.com/0/49275/ce23b661-b336-8b04-575c-9072052ab5c1.png)

コンパイルは通るのですが、下で Pharo が何かサジェスチョン「Use assert:equals: instead of assert: and = 」をしてくれていますのでレンチアイコンをクリックして従ってみましょう。

![Apply the proposed changes.png](https://qiita-image-store.s3.amazonaws.com/0/49275/b8cf3524-335c-43d2-358a-6d63245db21e.png)

いちおう、どんなふうに変えるか（赤部分を削って緑部分を追加）を教えてくれるみたいですね。問題なさそうなので OK します。

テストを実行します。

![800.png](https://qiita-image-store.s3.amazonaws.com/0/49275/b9757dfd-33a8-305d-e9b0-050b75921e8d.png)

Dollar>> * は定義されていませんから、もちろんレッドです。

イエローも仮実装もすっとばして一気にグリーンにします。Dollar>> = のときと同じように、Dollar クラスのプロトコルリスト枠にある -- all -- をクリックして次のコードをコンパイルしてメソッドを生成しましょう。

```Smalltalk
Dollar >> * multiplier
	^ (amount * multiplier) USD
```

コンパイルが終わったら、プロトコルリスト欄で 右クリック → Categorize all uncategorized もお忘れなく。（ * は arithmtic プロトコルに分類されます。）

![804.png](https://qiita-image-store.s3.amazonaws.com/0/49275/32b746de-d825-f5ad-b86a-d9c2a7ad3b88.png)

###### **TODO リスト**
- [ ] 5 USD + 10 CHF = 10 USD （レートが 2 : 1 の場合）
- [x] **5 USD * 2 = 10 USD**
- [x] 5 USD が Dollar を返す
- [x] 5 USD = 5 USD
- [x] 5 USD ~= 6 USD
- [x] インスタンス変数 amount を定義
- [x] amount のアクセッサーメソッド（amount 、amount: ） を追加
- [ ] hash の再定義
- [ ] printOn: の再定義


#ここまでのまとめと補足
- 等価チェックは = 、同一チェックは ==
- = のデフォルト動作（未定義時動作）は == 
- = を再定義したら hash も再定義（忘れていたのでそのうちやる）
- ; でメッセージを区切って連ねる「カスケード」という記法を使うと、テンポラリ変数を介してメッセージを畳みかけて送る処理と同等の記述ができる
- 式; yourself で、式のレシーバーを無条件に（式が何を返そうがそれを無視して）返すことができる
- メソッド名がよく利用されるものなら、Categorize all uncategorized で適切なプロトコルを推測して分類してくれる（でも全部は任せられないので、普段から調べて慣習を理解したり、自分で思いついたりできるようにしよう）
- 外部からはアクセッサーをコールするメッセージ式を介してしかインスタンス変数にはアクセスできない
- 引数をとるメソッドの名前はコロンを含み、コロンまで含めてひとつのメソッド名（例：amount: 、assert: 、printOn: 、assert:equals: 等々）でコロンなしの同名メソッドとは別物（例： amount と amount: 等）。さらに言うと同名のインスタンス変数とメソッド（アクセッサーメソッド）も別物
- インスタンスの文字列表現のデフォルトは 不定冠詞 + クラス名（an Object や a Dollar）なので、状態を持つようになったら printOn: を再定義して適切な表現に変える（忘れていたのでそのうちやる）

#この時点のコード
ファイルアウト形式のソースです。（この形式で Smalltalk のソースを読むことはあまり推奨されません。為念）

Pharo では、いったんこのコードをテキストファイルに保存してから、そのファイルを Pharo 内から Tools → File Browser で開き、Filein ボタンをクリックすると読み込めます。

Squeak でも読み込みや実行が可能なコードにもなっています。Pharo 同様に File List で開くか、デスクトップクリック → Workspace で開いたワークスペースなどにコピペし、alt/cmd + shift + g などで file it in するとクラスブラウザから閲覧したり、デスクトップクリック → Test Runner でテストも試せます（ただし使い勝手は Pharo に比べるとかなり落ちます^^;）

[[前回からの差分はこちら](https://github.com/sumim/tdd-money/commit/9b8fabf82b38c1d57ddfb1a6fad626641030a085)]

```Smalltalk
!Number methodsFor: '*TDD-Money' stamp: 'sumim 12/3/2017 17:27'!
USD
	^ Dollar new amount: self; yourself! !


Object subclass: #Dollar
	instanceVariableNames: 'amount'
	classVariableNames: ''
	poolDictionaries: ''
	category: 'TDD-Money'!


!Dollar methodsFor: 'accessing' stamp: 'sumim 12/3/2017 17:06'!
amount: aNumber
	amount := aNumber! !


!Dollar methodsFor: 'accessing' stamp: 'sumim 12/3/2017 17:06'!
amount
	^ amount! !


!Dollar methodsFor: 'comparing' stamp: 'sumim 12/3/2017 17:20'!
= other
	^ other class == self class and: [ other amount = amount ]! !


!Dollar methodsFor: 'arithmetic' stamp: 'sumim 12/3/2017 21:40'!
* multiplier
	^ (amount * multiplier) USD! !


TestCase subclass: #MoneyTest
	instanceVariableNames: ''
	classVariableNames: ''
	poolDictionaries: ''
	category: 'TDD-Money'!


!MoneyTest methodsFor: 'tests' stamp: 'sumim 12/3/2017 11:18'!
testFiveUSDReturnsADollar
    self assert: (5 USD isKindOf: Dollar)! !


!MoneyTest methodsFor: 'tests' stamp: 'sumim 12/3/2017 12:56'!
testEquality
	self assert: 5 USD = 5 USD.
	self assert: 5 USD ~= 6 USD! !


!MoneyTest methodsFor: 'tests' stamp: 'sumim 12/3/2017 17:45'!
testMultiplication
	self assert: 5 USD * 2 equals: 10 USD! !
```